#被动技能【黑棒】
execute as @a[tag=miyan,scores={skShangHai=1..,skPlayer=3..}] at @s run function nw:yingxiong/3yingxiong/10miyan/skill/1heibang
kill @e[type=trident,tag=heibang,scores={skTimeKe=50..}]

scoreboard players add @a[tag=miyan,scores={skShangHai=1..,skPlayer=0..3}] skPlayer 1
scoreboard players set @a[tag=miyan,scores={skShangHai=1..,skPlayer=4..}] skPlayer 0
scoreboard players reset @a[tag=miyan,scores={skShangHai=1..}] skShangHai
execute as @a[tag=miyan,scores={skPlayer=3}] at @s run effect give @s strength 999999 0 true
execute as @a[tag=miyan,scores={skPlayer=5}] at @s run effect give @s strength 999999 1 true
effect clear @a[tag=miyan,scores={skPlayer=..2}] strength


#主动技能【象转之术】触发
execute as @a[tag=miyan,scores={skEgg=1..}] at @s run function nw:yingxiong/3yingxiong/10miyan/skill/2xiangzhuan
#主动技能【象转之术】CD
execute as @a[tag=miyan,scores={skCDEgg=0}] at @s run function nw:yingxiong/3yingxiong/10miyan/skill/2xiangzhuan_cd
#主动技能【象转之术】效果
execute as @e[type=creeper,tag=xiangzhuan] at @s rotated ~ 0 run tp @s ^ ^ ^0.1
execute as @e[type=creeper,tag=xiangzhuan] at @s if entity @a[tag=!miyan,gamemode=!spectator,distance=0..3.5] run data merge entity @s {ignited:1b}
execute if entity @e[type=armor_stand,tag=wxty] as @e[type=creeper,tag=xiangzhuan] at @s facing entity @e[type=armor_stand,tag=wxty] feet run tp @s ^ ^0.2 ^5 ~ ~
kill @e[type=creeper,tag=xiangzhuan,scores={skTimeKe=600..}]

#主动技能【神罗天征】触发
execute at @a[tag=miyan,scores={skSneakTime=10..}] run particle firework ~ ~0.7 ~ 0.2 0.2 0.2 0.1 2 normal @a
execute as @a[tag=miyan,scores={skEXBottle=1..,skSneakTime=10..}] at @s run function nw:yingxiong/3yingxiong/10miyan/skill/3sltz
#主动技能【万象天引】触发
execute as @a[tag=miyan,scores={skEXBottle=1..,skSneakTime=..9}] at @s run function nw:yingxiong/3yingxiong/10miyan/skill/3wxty
#主动技能【万象天引】【神罗天征】CD
scoreboard players set @a[tag=miyan,scores={skSneak=0}] skSneakTime 0
scoreboard players set @a[tag=miyan,scores={skSneak=1..}] skSneak 0
execute as @a[tag=miyan,scores={skCDEXBottle=0}] run function nw:yingxiong/3yingxiong/10miyan/skill/3sltz_cd

#主动技能【万象天引】吸附效果
execute as @e[scores={miyan1=2..}] at @s unless block ~ ~ ~ #air run tp @s ~ ~0.5 ~
execute at @e[type=armor_stand,tag=wxty] run particle cloud ~ ~ ~ 0.2 0.2 0.2 0.01 5
execute as @e[scores={miyan1=2..}] at @s if entity @e[type=armor_stand,tag=wxty,distance=15..] facing entity @e[type=armor_stand,tag=wxty,scores={skTimeKe=2..},limit=1,sort=nearest] feet rotated ~ 0 run tp @s ^ ^0.05 ^3
execute as @e[scores={miyan1=2..}] at @s if entity @e[type=armor_stand,tag=wxty,distance=10..] facing entity @e[type=armor_stand,tag=wxty,scores={skTimeKe=2..},limit=1,sort=nearest] feet rotated ~ 0 run tp @s ^ ^0.05 ^2
execute as @e[scores={miyan1=2..}] at @s if entity @e[type=armor_stand,tag=wxty,distance=5..] facing entity @e[type=armor_stand,tag=wxty,scores={skTimeKe=2..},limit=1,sort=nearest] feet rotated ~ 0 run tp @s ^ ^0.05 ^1.2
execute as @e[scores={miyan1=2..}] at @s if entity @e[type=armor_stand,tag=wxty,distance=..5] facing entity @e[type=armor_stand,tag=wxty,scores={skTimeKe=2..},limit=1,sort=nearest] feet rotated ~ 0 run tp @s ^ ^0.05 ^0.5
scoreboard players reset @e[type=armor_stand,tag=wxty,scores={skTimeKe=2..}] skTimeKe

kill @e[type=armor_stand,tag=wxty,scores={skCDEntity=2..}]

#主动技能【神罗天征】击退效果
execute at @e[type=armor_stand,tag=sltz] run particle crit ~ ~ ~ 0.2 0.2 0.2 0.01 5
execute as @a[gamemode=!spectator,scores={miyan1=1}] at @s facing entity @e[type=armor_stand,tag=sltz,scores={skTimeKe=3..},limit=1,sort=nearest] feet rotated ~ 0 if block ^ ^1 ^-1 air if block ^ ^1 ^-2 air run tp @s ^ ^ ^-2
execute as @e[type=!player,type=!item,scores={miyan1=1}] at @s facing entity @e[type=armor_stand,tag=sltz,scores={skTimeKe=3..},limit=1,sort=nearest] feet rotated ~ 0 if block ^ ^1 ^-1 air if block ^ ^1 ^-2 air run tp @s ^ ^ ^-2
scoreboard players reset @e[type=armor_stand,tag=sltz,scores={skTimeKe=3..}] skTimeKe

#主动技能【神罗天征】撞墙效果
execute as @e[scores={miyan1=1}] at @s facing entity @e[type=armor_stand,tag=sltz,limit=1,sort=nearest] feet rotated ~ 0 unless block ^ ^1 ^-1 air run effect give @s instant_damage 1 0
execute as @e[scores={miyan1=1}] at @s facing entity @e[type=armor_stand,tag=sltz,limit=1,sort=nearest] feet rotated ~ 0 unless block ^ ^1 ^-1 air run scoreboard players set @s miyan1 0

execute if entity @e[type=armor_stand,tag=sltz,scores={skCDEntity=3..}] run scoreboard players set @e[scores={miyan1=1}] miyan1 0
kill @e[type=armor_stand,tag=sltz,scores={skCDEntity=3..}]

#10弥彦（天道-佩恩）
execute as @a[tag=miyan] run title @s actionbar [{"text":"黑棒","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skPlayer","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"/","color":"dark_red","bold":true,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"3","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 象转之术","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDEgg","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 万象天引/神罗天征","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDEXBottle","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]


#主动技能【地爆天星】

#主动技能【地爆天星】效果
